March 21, 2024.

Phugoid continues : linearization at a fixed point.

```
> R:='R':  # just in case it got changed.
  phug :=[D(theta)(t)=v(t)- cos(theta(t))/v(t), D(v)(t)=-sin(theta(t))-R*v(t)^2 ];
  xphug:=[op(phug), D(x)(t)=v(t)*cos(theta(t)), D(y)(t)=v(t)*sin(theta(t))]:
```

$$phug := \left[\mathrm{D}(\theta)(t) = v(t) - \frac{\cos(\theta(t))}{v(t)}, \mathrm{D}(v)(t) = -\sin(\theta(t)) - R\,v(t)^2\right] \quad (1)$$

Fix R to be some value, want to find a solution that doesn't change in v and θ, ie, a fixed point.;
That is, the vector field is zero there in both components.

```
> subs({theta(t)=theta, v(t)=v}, phug);
```

$$\left[\mathrm{D}(\theta)(t) = v - \frac{\cos(\theta)}{v}, \mathrm{D}(v)(t) = -\sin(\theta) - R\,v^2\right] \quad (2)$$

I don't wan't the D(theta)(t) etc, either... that should be 0. I will use map to do this for me.

```
> map(eq->rhs(eq)=0, %);
```

$$\left[v - \frac{\cos(\theta)}{v} = 0, -\sin(\theta) - R\,v^2 = 0\right] \quad (3)$$

```
> solve( %, {v,theta});
```

$$\{\theta = \arctan(-RootOf(-1 + (R^2+1)\_Z^2)\,R, RootOf(-1 + (R^2+1)\_Z^2)), v = RootOf(\_Z^2 - RootOf(-1 + (R^2+1)\_Z^2))\} \quad (4)$$

That's ugly, let's clean it up into a more readable form.

```
> convert(%,radical);
```

$$\left\{\theta = \arctan\left(-\sqrt{\frac{1}{R^2+1}}\,R, \sqrt{\frac{1}{R^2+1}}\right), v = \left(\frac{1}{R^2+1}\right)^{1/4}\right\} \quad (5)$$

Redo the whole thing in one command:

```
> Sol:=convert(
   solve(
     map(eq->rhs(eq)=0,
      subs({theta(t)=theta, v(t)=v}, phug)),
   {v,theta}), radical);
```

$$Sol := \left\{\theta = \arctan\left(-\sqrt{\frac{1}{R^2+1}}\,R, \sqrt{\frac{1}{R^2+1}}\right), v = \left(\frac{1}{R^2+1}\right)^{1/4}\right\} \quad (6)$$

I want a function Sol(R) that gives me $[\theta(R), v(R)]$ at the fixed point, so let's modify this a bit:

```
> Sol:=proc(Rval)
   local mypair;
   mypair:=
    convert(
      solve(
       map(eq->rhs(eq)=0,
          subs({theta(t)=theta, v(t)=v}, phug)),
      {v,theta}),
    radical);
   return(eval(mypair,R=Rval));
```

```
  end:
> Sol(R);
```

$$\left\{\theta = \arctan\left(-\sqrt{\frac{1}{R^2+1}}\,R, \sqrt{\frac{1}{R^2+1}}\right), v = \left(\frac{1}{R^2+1}\right)^{1/4}\right\} \qquad (7)$$

I could have used subs, but eval is kind of better, since it gives me a number if Rval is a number, rather than something involving arctan.

```
> Sol(.25)
```

$$\{\theta = -0.2449786631, v = 0.9849581210\} \qquad (8)$$

Now we just want to transform this to a pair of values $[\theta, v]$

```
> Sol:=proc(Rval)
   local mypair,eqns;
   mypair:=
    convert(
      solve(
       map(eq->rhs(eq)=0,
          subs({theta(t)=theta, v(t)=v}, phug)),
      {v,theta}),
    radical);
   eqns:=eval(mypair,R=Rval);
   return(subs(eqns,[theta,v]));
  end:
> Sol(R);
```

$$\left[\arctan\left(-\sqrt{\frac{1}{R^2+1}}\,R, \sqrt{\frac{1}{R^2+1}}\right), \left(\frac{1}{R^2+1}\right)^{1/4}\right] \qquad (9)$$

```
> Sol(0.25);
```

$$[-0.2449786631, 0.9849581210] \qquad (10)$$

My goal in all of this was to look at how the fixed point varies with R

```
> plot(Sol(R),R=0..3,legend=[theta,v], size=[.5,"golden"]);
```



This works, but not really what I wanted. I want a parametric plot of where the fixed point is as a function of R.

```
> plot([Sol(R),R=0..3]);
Error, (in plot) incorrect first argument [[arctan(-(1/(R^2+1))^(1/2)*R,(1/(R^2+1))^(1/2)), (1/(R^2+1))^(1/4)], R = 0 .. 3]
```

Duh... we want a triple , so I have to "open up" the answer and insert the R range inside it.

```
> plot([op(Sol(R)),R=0..5],  theta=-Pi/2..0.1, v=0..1.2,
```

```
tickmarks=[piticks,default],size=[.6,.6],
title="location of fixed point for 0<R<5");
```



Move on to linear diff eqn's. Some discussion at the board.

```
> lin:= [ D(x)(t)=a*x(t), D(y)(t)=d*y(t) ];
```

$$lin := [\mathrm{D}(x)(t) = a\,x(t), \mathrm{D}(y)(t) = d\,y(t)] \quad (11)$$

```
> dsolve(lin);
```

$$\{x(t) = c_2\,\mathrm{e}^{a\,t}, y(t) = c_1\,\mathrm{e}^{d\,t}\} \quad (12)$$

```
> with(DEtools):
```

$$a := 2$$
$$d := 2 \quad (13)$$

```
> a:=2; d:=2;
DEplot(lin, [x(t),y(t)], t=-1..1, [seq(seq([x(0)=x0,y(0)=y0],x0=
-1..1,.4),y0=-1..1,.4)]);
```

$$a := 2$$
$$d := 2$$




```
> a:=1; d:=3;
  DEplot(lin, [x(t),y(t)], t=-1..1, [seq(seq([x(0)=x0,y(0)=y0],x0=
  -1..1,.4),y0=-1..1,.4)]);
```

$$a := 1$$

$$d := 3$$




```
> a:=-1; d:=2;
  DEplot(lin, [x(t),y(t)], t=-1..1, [seq(seq([x(0)=x0,y(0)=y0],x0=
  -1..1,.4),y0=-1..1,.4)]);
```

$$a := -1$$

$$d := 2$$




```
> b:=-1; c:=2;
  DEplot([ D(x)(t)=b*y(t), D(y)(t)=c*x(t) ], [y(t),x(t)], t=-1..1,
  [seq(seq([x(0)=x0,y(0)=y0],x0=-1..1,.4),y0=-1..1,.4)], scaling=
  constrained, axes=boxed);
```

$$b := -1$$

$$c := 2$$




```
> a:=.5; b:=-1; c:=2;
  DEplot([ D(x)(t)=a*x(t)+b*y(t), D(y)(t)=c*x(t) ], [y(t),x(t)], t=
  -1..1, [seq(seq([x(0)=x0,y(0)=y0],x0=-1..1,.4),y0=-1..1,.4)],
  scaling=constrained, axes=boxed);
```

$$a := 0.5$$

$$b := -1$$

$$c := 2$$




```
> a:=-1; b:=-1; c:=2;
  DEplot([ D(x)(t)=a*x(t)+b*y(t), D(y)(t)=c*x(t) ], [y(t),x(t)], t=
  -1..1, [seq(seq([x(0)=x0,y(0)=y0],x0=-1..1,.4),y0=-1..1,.4)],
  scaling=constrained, axes=boxed);
```

$$a := -1$$

$$b := -1$$

$$c := 2$$




```
> a:=-2; b:=-1; c:=2;
  DEplot([ D(x)(t)=a*x(t)+b*y(t), D(y)(t)=c*x(t) ], [y(t),x(t)], t=
  -1..1, [seq(seq([x(0)=x0,y(0)=y0],x0=-1..1,.4),y0=-1..1,.4)],
  scaling=constrained, axes=boxed);
```

$$a := -2$$

$$b := -1$$

$$c := 2$$


6
4
2
0
−2
−4
−6
x
−4
−2
0
2
4
y